24. Тип файл

Классификация: нестандартный, структурированный (сложный) тип.

Имя определяет программист.

3. Структурная организация

Файл - последовательность однотипных компонентов, ограниченная специальным признаком - конец файла, расположенная на внешнем устройстве. Одновременно с файлом в оперативной памяти выделяется участок, по размеру совпадающий с одним компонентом файла, называемый буфером или окном файла. Вся работа с файлом производится через этот буфер.

По структуре - файл есть последовательность элементов. Принято компонент файла называть записью (это сложилось исторически до появления языка Паскаль). Необходимо отличать понятия запись в файле от записи, как типа данных в Паскале. Запись в файле - достигнутый компонент при выполнении операции чтении из файла или операции записи в файл. Размер записи (компонента файла) в файле определяется типом компонента файла. 

При определении переменной типа файл мы определяем две структуры - структуру на внешнем устройстве и структуру в оперативной памяти (рис. 24.1) 

Структура на внешнем устройстве представляет собой последовательность однотипных компонентов, ограниченную признаком конец файла. Для связи переменной типа файл с конкретным внешним устройством задаются специальные команды. 

Структура в оперативной памяти - буфер или окно файла, размером в один компонент файла. Все общение с файлом, расположенном на внешнем устройстве, осуществляется через буфер файла, который находится в оперативной памяти.

Рис. 24.1. – Структурная организация данных типа файл

0. Определение типа

Для того, чтобы определить тип файл необходимо определить тип элементов, которые могут входить в файл. Количество компонентов не определяется. Типом компонентов может быть любой, кроме файлового, тип. Непосредственное определение типа файл приведено на рис. 24.2.

Рис. 24.2. – Определение типа файл

Как всегда, это определение делается в разделе нестандартных типов. Однако язык позволяет не определять тип компонента файла. В этом случае в качестве типа данных указывается тип FILE (такой файловый тип называется нетипизированным). Для нетипизированного файла нельзя выполнять стандартные операции по чтению и записи.

Примеры определения файловых типов и файлов:

TYPE


tf1=file of integer;


tf2=file of real;

VAR


f1:tf1;


f2:tf2;

f3:FILE;

В этих примерах f1 – файл из целых чисел, f2 – файл из вещественных чисел, f3 - нетипизированный файл. С файлом f3 нельзя выполнять стандартные операции по чтению и записи.

1. Множество значений

Включает в себя все значения, которые могут быть размещены в структуре, заданной определением типа.

2. Множество операций

Все операции с переменными типа файл (в дальнейшем будем говорить - с файлом) сводятся к двум: запись в файл и чтение из файла. Правила выполнения этих операций зависят от вида файлов. Виды файлов определяются техническими возможностями внешних устройств, на которых располагаются файлы.

Физические и логические файлы

Для независимости программы обработки файлов от конкретной операционной системы один и тот же обрабатываемый файл одновременно является физическим и логическим. 

Логический файл – это имя переменной файлового типа в программе. Выполнение операторов программы по обработке файла осуществляется посредством этой переменной. Переменная файлового типа в программе называется именем логического файла.

Физический файл – это файл, определенный по правилам конкретной операционной системы. В любой операционной системе файл определяется как поименованная порция информации, расположенная на внешнем устройстве. Это имя называется именем физического файла. 

Во всех операционных системах, применяемых для ПЭВМ типа IBM, не дисковые файлы именуются именами внешних устройств, на которых они располагаются. Обязательные внешние устройства имеют свои уникальные имена (CON – клавиатура при чтении-вводе и экран дисплея при записи-выводе, PRN – принтер и др.). Дополнительные (факультативные) внешние устройства именуются именами портов ввода-вывода, к которым они подключены (LPT1, LPT2, LPT3, COM1, COM2, COM3 и т.д.). Так как обычно принтер подключается к первому параллельному порту, имена PRN и LPT1 являются синонимами.

Имена дисковых файлов состоят из имен дисковода, на котором они расположены, имен каталогов-папок, в которые они вложены и из полных имен файлов (составные части разделяются обратными слэшами \). Например, С:\TP60\WORK\dan.dat – имя физического файла, расположенного на диске С:, вложенного внутрь каталога ТР60, подкаталолга WORK, полное имя этого файла dan.dat. 

Для того чтобы программа обрабатывала реальный файл, до начала обработки файла обязательно логическому файлу в программе сопоставляется физический файл в операционной системе. Это называется связыванием логического и физического файлов. 

Связывание осуществляется с помощью стандартной процедуры assign, имеющей два параметра. Первый параметр – имя логического файла. Второй параметр – строка символов с именем физического файла, заданного по правилам конкретной операционной системы.

Виды файлов

Различают три вида файлов:

· последовательные;

· прямого доступа;

· текстовые или символьные.

Эти виды определяют способы организации работы с файлами. 

Разбиение на виды находится в непосредственной зависимости от технических возможностей устройств, на которых размещаются файлы. Так на экране дисплея может размещаться только текстовый файл, файл такого же типа размещается на клавиатуре ПЭВМ. На магнитофоне может размещаться только последовательный файл. На магнитном диске может размещаться файл любого вида, но наибольший интерес здесь представляют файлы прямого доступа, так как механизм перемещения головок дисковода над поверхностью диска обеспечивает доступ к любому компоненту файла, в противоположность механизму магнитофона, который требует прокрутить ленту (просмотреть все предыдущие компоненты), чтобы выбрать необходимый элемент.

Общие правила работы с файлами

При работе с файлами любого вида необходимо придерживаться некоторых общих правил:

a) в разделе нестандартных типов определяется файловый тип (для нетипизированных файлов это делать не надо, так как для указания этого типа используется имя FILE);

б) файл описывается в разделе описания переменных путем указания имени файловой переменной и типа файла (таким образом определяется имя логического файла);

в) в разделе операторов до использования файла файловая переменная связывается с конкретным физическим файлом:

assign (<имя файловой переменной, называемое именем логического файла>, '<имя физического файла>');

г) перед выполнением операций с файлом, он открывается для использования с помощью одной из команд (обращения к процедуре):

· reset(<имя файловой переменной>) - открытие существующего файла для работы. После выполнения этой команды буфер файла связывается с первым компонентом файла (т.е. в буфере располагается содержимое первого компонента файла) и можно начинать работу с файлом. Попытка открыть с помощью этой процедуры несуществующий файл приводит к ошибке ввода-вывода, в этом случае при включенной системе прерываний по вводу-выводу программа снимается с решения;

· rewrite(<имя файловой переменной>) - открытие для формирования (записи) нового файла. После выполнения этой команды в файле первым (и единственным) компонентом будет признак конца файла, значение буфера файла неопределенное, и буфер связан с этим первым компонентом, и можно начинать работу с файлом. Необходимо помнить, что если до выполнения этой процедуры файл существовал, то вся информация из него "пропадает", так как в начало сразу же пишется признак конец файла;

·  (только для текстовых файлов!) append(<имя файловой переменной типа text>) – открытие для дозаписи в конец существующего текстового файла. После выполнения этой команды буфер файла связывается с последним компонентом файла - признаком конец файла, значение буфера файла станет неопределенным, и можно начинать работу с файлом. Под работой подразумевается последовательная запись символов в конец файла. Если до выполнения процедуры файл не существовал, то возникает ошибка ввода-вывода;

При открытии файла, он монополизируется, т.е. становится доступным только данной программе и недоступным всем другим программам;

д)начинается непосредственная работа с файлом по чтению и записью Она строго определяется видом файла. При этом обязательно используются стандартные процедуры чтения и записи READ и WRITE. Для текстовых файлов добавляются процедуры READLN и WRITELN;

е)после окончания использования, файл должен быть закрыт с помощью обращения к процедуре close(<имя файловой переменной>). При выполнении этой процедуры ожидается завершение внешним устройством всех запланированных операций чтения и записи, после чего файл демонополизируется, т.е. становится недоступным данной программе и доступным всем остальным.

Необходимо помнить, что до тех пор, пока файл открыт в одной программе, он не может быть использован в другой. Если в программе файл не закрывается, то после окончания работы программы он закроется автоматически, при этом из-за асинхронности работы центрального процессора и контроллера, управляющего работой внешнего устройства, последняя операция по чтению-записи может быть не завершена (критично, когда осуществляется операция записи).

24.4. Организация работы с файлами прямого доступа

Файлы прямого доступа характерны тем, что в любой момент времени можно получить доступ к любому существующему компоненту файла - связать буфер файла с любым существующим компонентом файла.

Все компоненты файла прямого доступа автоматически пронумерованы, начиная с 0. Номер компонента файла типа longint, т.е. теоретически в файле прямого доступа может быть свыше 2 млн. компонентов (правда, пока таких внешних устройств не существует). 

В файлах прямого доступа читать можно любой существующий компонент файла, а записывать в файл можно  либо на место любого существующего компонента, либо в конец файла.

С файлами прямого доступа можно работать в двух режимах:

· работа с существующим файлом прямого доступа. Открывается режим с помощью процедуры reset(<имя логического файла>). Попытка открыть этот режим для несуществующего файла приводит к ошибке ввода-вывода;

· создание нового файла прямого доступа . Открывается режим с помощью процедуры rewrite(<имя логического файла>). Если до открытия в этом режиме файл существовал, то информация в нем уничтожается.

При работе с файлами прямого доступа можно использовать следующие процедуры и функции (во всех нижеприведенных подпрограммах f – это имя логического файла):

· procedure write(var f:<имя файлового типа>; r:<тип компонента файла>) - процедура записи в текущий компонент файла значения выражения, указанного при обращении к процедуре. После выполнения процедуры записи буфер файла связывается со следующим по порядку компонентом файла. Если до выполнения процедуры текущим компонентом был конец файла, то, после выполнения записи на место этого признака, в следующий компонент файла запишется признак конца файла, и буфер свяжется с этим признаком.

· procedure read(var f:<имя файлового типа>;var r:<тип компонента файла>) - процедура чтения текущего компонента файла. Прочитанное из текущего компонента значение помещается в переменную, указанную при обращении к процедуре. Текущий компонент файла - это тот компонент, с которым связан буфер файла. После выполнения процедуры чтения буфер файла связывается со следующим по порядку компонентом файла. Выполнение процедуры возможно только в том случае, если текущий компонент не является признаком конец файла. Если буфер связан с этим признаком, то при выполнении процедуры происходит ошибка ввода-вывода;

· function eof(var f:<имя файлового типа>):Boolean – с помощью этой функции можно проверить связан ли буфер с признаком конец файла. Если результат true, то буфер связан с этим признаком, если – false, то буфер не связан с признаком конец файла, а находится на каком-то существующем компоненте файла;

· procedure seek(var f:<имя файлового типа>; n:longint) - процедура установки на n-ый компонент файла прямого доступа f. При выполнении процедуры буфер файла связывается с n-ым компонентом, если такой компонент существует. В случае, если файл содержит меньшее количество компонентов, произойдет ошибка ввода-вывода ( в этом случае при включенной системе прерываний по вводу-выводу программа снимется с решения);

· procedure reset(var f:<имя файлового типа>) - процедура установки на начало файла - буфер связывается с первым компонентом файла. Обращения seek(f,0) и reset(f) в процессе работы с файлом прямого доступа эквивалентны;

· procedure truncate(var f:<имя файлового типа>) - усекает файл, начиная с текущей позиции. После выполнения процедуры в текущую позицию, с которой был связан буфер, будет записан признак конца файла;

· function filepos(var f:<имя файлового типа>):longint - функция определения текущей позиции файла;

· function filesize(var f:<имя файлового типа>):longint -функция определения количества компонентов в файле (признак конца файла не учитывается).

Алгоритм сортировки файла прямого доступа

Сортировка файла методом пузырька (обменная сортировка)

Спецификация подпрограммы

1. Назначение: сортировка файла прямого доступа из целых чисел по убыванию (файл обязательно существует; количество компонентов в файле произвольное).Файловый тип определяется ранее подпрограммы следующим образом 
type tf=file of integer;

2. Имя: sortfilepuz
3. Вид: процедура

4. Перечень параметров:

Таблица 24.1. Перечень параметров

	Статус
	Назначение
	Имя
	Тип
	Вид

	Вход
	имя физического файла
	name
	string
	параметр-значение


5. Заголовок подпрограммы: procedure sortpuz(name:string);
Метод решения

Так как в файле прямого доступа все существующие компоненты пронумерованы (то есть проиндексированы), то для сортировки файла прямого доступа можно использовать любой метод сортировки массива. Используем обменную сортировку с флагом, известную под названием метод "пузырька". При  этом для сокращения количества переборов будем просматривать только неупорядоченную часть.

1. Связываем логический и физический файлы

assign(f,name)
;

2. Открываем для работы файл прямого доступа в режиме работы с существующим файлом

reset(f)
;

3. Определяем размер не отсортированной части файла (перед началом сортировки это все существующие компоненты файла)

nkz:=filesize(f)
;

4. Повторяем просмотр не отсортированной части элементов до выполнения условия до выполнения условия – при очередном переборе элементов перестановки не было (признак  отсутствия перестановки – флаг flag при отсутствии перестановки имеет значение ложь, а при наличии – значение истина)

повторять

  <просмотр не отсортированной части файла>

до 

При просмотре не отсортированной части выполняются следующие действия:

а) признаку перестановки при начале очередного перебора присваивается значение ложь (нет перестановки)

flag:=false
;

b) перебираются все элементы не отсортированной части c начального по предпоследний. Начальный элемент имеет порядковый номер 0; предпоследний элемент – nkz-2 (минус два, т.к. нумерация не с единицы, а с нуля)

Для каждого номера элемента i выполняются следующие действия

b.1) буфер файла связывается с текущим элементом

seek(f,i)
;

b.2) читаются текущий и следующий элементы файла

read(f,r1,r2)
.

Компилятор, обрабатывая такое обращение к процедуре чтения из файла, преобразует его в эквивалентный составной оператор

begin

  read(f,r1);

  read(f,r2)

end;

b.3) при сортировки по убыванию, если следующий элемент больше текущего (при сортировки по возрастанию, если меньше), то буфер файла связывается с текущим компонентом; записывается в файл следующее значение, а затем текущее; формируется признак наличия перестановки (флагу перестановки присваивается истина)

Компилятор, обрабатывая обращение к процедуре записи в файл, преобразует его в эквивалентный составной оператор

begin

  write(f,r2);

  write(f,r1)

end;

c) так как в правую часть файла переместился отсортированный элемент, то размер не отсортированной части уменьшается на единицу

nkz:=nkz-1
;

5. Закрываем файл для работы

close(f)
.

Информационная модель

Таблица 24.2. Информационная модель

	Назначение
	Имя
	Тип

	Имя логического файла
	f
	tf

	Номер очередного компонента файла
	i
	longint

	Размер неотсортированной части файла
	nkz
	longint

	Значение очередного компонента файла
	r1
	integer

	Значение следующего компонента файла
	r2
	integer

	Признак наличия перестановки
	flag
	boolean


Программная модель

type tf=file of integer;

procedure sortpuz(name:string);

var f:tf;

    i,nkz:longint;

    r1,r2:integer;

    flag:boolean;

begin

  assign(f,name);

  reset(f);

  nkz:=filesize(f);

  repeat

    flag:=false;

    for i:=0 to nkz-2 do

      begin

        seek(f,i);

        read(f,r1,r2);

        if r2>r1 then

          begin

            seek(f,i);

            write(f,r2,r1);

            flag:=true

          end

      end;

  nkz:=nkz-1

  until not flag;

  close(f);

end;

24.5. Организация работы с последовательными файлами

Под последовательными файлами понимают такие файлы, в которых искать необходимую информацию можно только последовательно читая компоненты файла, начиная с начала файла, а записывать новые компоненты только в конец файла. Причем из файла можно либо только читать информацию, либо только записывать информацию.

В последовательных файлах не существует понятия номер компонента файла. Поэтому при работе с последовательными файлами можно использовать все процедуры и функции, определенные для файлов прямого доступа, за исключением процедуры установки на компонент файла seek, функций, связанных с определением номера и количества компонентов filesize и filepos, процедуры усечения файла truncate.

24.6. Разработка комплекса программ СУБД 
Комплекс программ системы управления базами данных предназначен для создания, ведения и использования баз данных пользователями. Программы – это реализация функций, возложенных на СУБД. В простейшем случае база данных хранится в виде файла прямого доступа. В качестве примера базы данных возьмем базу данных с информацией об учащихся:

· фамилия (строка из 20 символов);

· дата рождения (день, месяц, год);

· год поступления в учебное учреждение (целое без знака);

· средний балл за период обучения (вещественное число).

База данных хранится в файле прямого доступа, компонентами файла являются данные типа запись:

type tdata=record

       d:1..31;

       m:1..12;

       g:word

     end;

{тип компонента файла}

     tinfstud=record

       fam:string[20];

       dr:tdata;

       gp:word;

       sb:real

     end;

{файловый тип}

     tfileinfstud=file of tinfstud;

Реализуем следующие функции базы данных:

· создание БД (создание файла прямого доступа);

· просмотр БД на экране (вывод на экран содержимого файла прямого доступа);

· сортировка БД (сортировка записей файла прямого доступа по какому-то полю записи);

· выборка из БД (поиск и выборка информации из файла с БД и создание нового файла прямого доступа с выбранными записями).

Обычно каждая функция БД реализуется в виде отдельной программы, выбор необходимой пользователю функции реализуется с помощью программы-меню. В данном примере реализуем только функции БД в виде отдельных программ и определим технологию работы с ними. Все описания поместим в отдельный текстовый файл (дадим ему имя FILE0.PAS), который будем подключать к тексту каждой программы с помощью директивы компилятору – включить текстовый файл {$I <имя файла>}. В этом же текстовом файле разместим подпрограмму проверки существования файла, так как эта подпрограмма потребуется во всех программах комплекса.

Разработка подпрограммы для проверки существования файла

Спецификация

1. Назначение: проверка существования физического файла вне зависимости от его вида

2. Имя: FS
3. Вид: процедура

4. Перечень параметров:

Таблица 24.3. Перечень параметров

	Статус
	Назначение
	Имя
	Тип
	Вид

	Вход
	имя физического файла
	name
	string
	параметр-значение

	Возвращ.
	признак существования файла (true –файл существует, false – нет)
	fs
	Boolean
	-


5. Заголовок подпрограммы:  function fs(name:string):Boolean;
Метод решения

В основу метода решения положено условие выполнения процедуры открытия существующего файла для работы reset(<илф>). Если физический файл соответствующий заданному логическому существует, то файл открывается для работы, если отсутствует, то возникает ошибка ввода-вывода. Поэтому метод решения следующий:

1) связываем физический и логический файлы assign(f, name), где f – имя логического нетипизированного файла. Нетипизированный файл позволяет сделать подпрограмму независимой от вида файла (но с таким файлом нельзя выполнять стандартные операции чтения и записи);

2) отключаем систему прерываний по ошибкам ввода-вывода {$I-};

3) выполняем вызов процедуры reset(f);
4) включаем систему прерываний по ошибкам ввода-вывода {$I+};

5) анализируем наличие ошибки при выполнении последней операции ввода-вывода. Это может быть только одна ошибка – перед выполнением процедуры reset физический файл, соответствующий логическому отсутствует. Если ошибка есть, то формируем результат функции false. Если ошибки нет, то формируем результат функции true и демонополизируем файл

Информационная модель

Таблица 24.4. Информационная модель

	Назначение
	Имя
	Тип

	Имя логического файла
	f
	file


Текст подпрограммы

function fs(name:string):boolean;

var f:file;

begin

  assign(f,name);

  {$I-}

  reset(f);

  {$I+}

  if ioresult<>0 then

    fs:=false

  else

    begin

      fs:=true;

      close(f)

    end
end;

Реализация функции создания БД

Постановка задачи 

Создать файл прямого доступа с информацией об учащихся. Имя создаваемого физического файла определяет пользователь. Компоненты файла представляют собой записи типа tinfstud.

Метод решения

1) получить от пользователя имя физического файла (namebd) для хранения базы данных;

2) если файл с таким именем существует, то сообщить об этом пользователю и прекратить выполнение программы

3) связать логический и физический файлы assign(fbd,namebd);
4) открыть файл прямого доступа в режиме создания нового файла
rewrite(fbd);
5) повторять до ответа пользователя не продолжать создание файла:
a) ввод информации об очередном студенте (ввод записи r);

b) запись информации в файл write(fbd,r);

c) получение ответа на запрос – продолжать создание базы данных?[Y/N]

6) закрыть файл для работы close(fbd).
Предложенный метод можно использовать и для создания последовательных файлов, так как никаких особенностей файлов прямого доступа не используется.

Информационная модель

Таблица 24.5. Информационная модель

	Статус
	Назначение
	Имя
	Тип

	Вход
	имя физического файла с БД
	namebd
	string

	Пром
	имя логического файла с БД
	fbd
	tfileinfstud

	Пром
	информация об очередном учащемся
	r
	tinfstud

	Вход
	ответ на запрос(Y – продолжать, N – нет)
	otvet
	char


Программная модель

{программа создания базы данных (файл Z1.pas)}

program z1;

{$I FILE0.PAS}

var otvet:char;

begin

  writeln('Введите имя создаваемого файла с базой данных');

  readln(namebd);

  if fs(namebd) then

    begin

      writeln('ОШИБКА!!! Файл с именем ',namebd,' уже существует');

      halt

    end;

  assign(fbd,namebd);

  rewrite(fbd);

  repeat
    writeln('Введите информацию об очередном студенте?');

    with r do

      begin

        writeln('Фамилия?');

        readln(fam);

        writeln('День, месяц, год рождения, год поступления, средний балл ',

                 'для студента ',fam);

        readln(dr.d, dr.m, dr.g, gp, sb);

        write(fbd,r)

      end;

    writeln('Продолжать?[y/n/]');

    readln(otvet)

  until (upcase(otvet)='N') or (otvet='т') or (otvet='Т');

  close(fbd)

end.

Реализация функции просмотра БД

Постановка задачи 

Просмотр на экране содержимого файла типа tfileinfstud. Имя физического файла для просмотра задает пользователь.

Метод решения

При выводе информации на экран будем использовать алгоритм вывода большого количества значений.

1) получить от пользователя имя физического файла, который будем просматривать readln(namebd); 

2) если физический файл с таким именем отсутствует, то сообщить об этом пользователю и прекратить выполнение программы

3) связать логический и физический файлы assign(fbd,namebd);
4) открыть файл в режиме работы с существующим файлом (в данном методе будем только последовательно читать информацию из файла) reset(fbd);
5) очистить экран clrscr;
6) вывести шапку таблицы 
writeln('ФАМИЛИЯ         ':20,' ДАТА РОЖД.','  ГП ',' БАЛЛ');
7) в счетчик заполненных строк экрана занести 1 k:=1;

8) пока не достигли признака конца в файле fbd повторять (not eof(fbd))

a) чтение очередного компонента файла read(fbd,r);

b) вывод на экран компонентов записи r с информацией об учащемся;

c) увеличение счетчика выведенных строк на 1 k:=k+1;

d) 

9) закрыть файл для работы close(fbd).

Предложенный метод решения позволяет выводить на экран содержимое любого файла типа tfileinfstud. Причем файл может быть прямого доступа, так и последовательным, потому что в методе решения не используются ни какие особенности файлов прямого доступа.

Информационная модель

Таблица 24.6. Информационная модель

	Статус
	Назначение
	Имя
	Тип

	Вход
	имя физического файла с компонентами типа tinfstud
	namebd
	string

	Пром
	имя логического файла 
	fbd
	tfileinfstud

	Пром
	информация об очередном учащемся
	r
	tinfstud

	Вход
	счетчик заполненных строк экрана
	k
	integer


Программная модель

{программа просмотра файлов с компонентами типа tinfstud (файл z2.pas)}

program z2;

uses CRT;

{$I FILE0.PAS}

var k:integer;

begin

  writeln('Введите имя просматриваемого файла ');

  readln(namebd);

  if not fs(namebd) then

    begin

      writeln('ОШИБКА!!! Файл с именем ',namebd,' не существует');

      halt

    end;

  assign(fbd,namebd);

  reset(fbd);

  clrscr;

  writeln('ФАМИЛИЯ         ':20,' ДАТА РОЖД.','  ГП ',' БАЛЛ');

  k:=1;

  while not eof(fbd) do

    begin

      read(fbd,r);

      with r do

        begin

          fam:=copy(fam+'                         ',1,20);

          writeln(fam:20, dr.d:3,'.',dr.m:2,'.', dr.g:4, gp:5, sb:5:2)

          k:=k+1;

          if k=23 then

             begin
               writeln('Для продолжения нажмите Enter');

               readln;

               clrscr;

               writeln('ФАМИЛИЯ         ':20,' ДАТА РОЖД.',

                       '  ГП ',' БАЛЛ');

               k:=1;

             end

        end

    end;

  close(fbd);

  writeln('Информация исчерпана!!!')

end.

Реализация функции сортировки БД

Постановка задачи 

Сортировка файла прямого доступа с информацией об учащихся. Фамилии упорядочиваются по алфавиту, если фамилии совпадают, то информация об однофамильцах упорядочивается по возрастанию дат рождения. Имя сортируемого физического файла определяет пользователь. Компоненты файла представляют собой записи типа tinfstud.

Метод решения

Так как файл прямого доступа в полной мере является аналогом массива, то для сортировки можно использовать любой метод сортировки массива. Выбираем метод сортировки выбором с перестановкой:

1) получить от пользователя имя физического файла, который будем сортировать readln(namebd); 

2) если физический файл с таким именем отсутствует, то сообщить об этом пользователю и прекратить выполнение программы

3) связать логический и физический файлы assign(fbd,namebd);

4) формируем все компоненты файла с начального по предпоследний. В файлах прямого доступа начальный компонент имеет номер 0. Количество существующих компонентов файла прямого доступа определяется с помощью функции filesize(<илф>), так как нумерация с 0, то номер предпоследнего компонента файла на 2 меньше, чем количество компонентов
i:=0..filesize(fbd)-2:
5) для каждого формируемого компонента i:

a) выбираем экстремальный компонент, начиная с формируемого компонента по последний элемент файла и определяем его местоположение;

b) осуществляем перестановку значений с формируемого места и места экстремального значения;

6) закрываем файл для работы close(fbd).

a) Выбор экстремального компонента осуществляем следующим образом:

1) читаем из файла очередной компонент (он находится на формируемом месте i) и принимаем его значение за экстремальное read(fbd,extr);

2) фиксируем место экстремального значения nm:=i;

3) перебираем все остальные компоненты файла. Пока не достигли конца файла  not eof(fbd) повторяем

· чтение очередного компонента файла в переменную r read(fbd,r);

· если (r.fam<extr.fam) 

b) Перестановка значений

Перед выполнением этой подзадачи сложилась ситуация показанная на рис.24.3

Рис. 24.3. – Результат выбора

1 шаг перестановки – на место экстремального значения помещается значение с формируемого места:

· позиционирование на компонент i  seek(fbd,i);

· .чтение из файла в промежуточную переменную r read(fbd,r);
· позиционирование на компонент nm seek(fbd,nm);
· запись значения переменной r в файл write(fbd,r);
2 шаг перестановки – на формируемое место записывается экстремальное значение

· позиционирование на компонент i  seek(fbd,i);

· запись значения переменной extr в файл write(fbd,extr);

Обратите внимание, что после выполнения действий по перестановке, буфер связан с i+1 компонентом, то есть для выполнения п.5. буфер файла связан с необходимым компонентом.

Информационная модель

Таблица 24.7. Информационная модель

	Статус
	Назначение
	Имя
	Тип

	Вход
	имя физического файла с компонентами типа tinfstud
	namebd
	string

	Пром
	имя логического файла 
	fbd
	tfileinfstud

	Пром
	номер формируемого компонента файла
	i
	longint

	Пром
	экстремальное значение
	extr
	tinfstud

	Пром
	номер компонента с экстремальным значением
	nm
	longint

	Пром
	информация об учащемся
	r
	tinfstud


Программная модель

{программа сортировки (файл Z3.PAS)}

program z3;

{$I FILE0.PAS}

var i,nm:longint;

    extr:tinfstud;

begin
  writeln('Введите имя файла с информацией о студентах');

  readln(namebd);

  if not fs(namebd) then

    begin

      writeln('ОШИБКА!!! Файл с именем ',namebd,' не существует');

      halt

    end;

  assign(fbd,namebd);

  reset(fbd);

  for i:=0 to filesize(fbd)-2 do

    begin

      read(fbd,extr);

      nm:=i;

      while not eof(fbd) do

        begin

          read(fbd,r);

          with r do

            if (fam<extr.fam)or

               (fam=extr.fam)and((dr.g<extr.dr.g)or

               (dr.g=extr.dr.g)and((dr.m<extr.dr.m)or

               (dr.m=extr.dr.m)and(dr.d<extr.dr.d)))

            then

              begin

                extr:=r;

                nm:=filepos(fbd)-1

              end

          end;

       seek(fbd,i);

       read(fbd,r);

       seek(fbd,nm);

       write(fbd,r);

       seek(fbd,i);

       write(fbd,extr)

    end;

  close(fbd);

end.

Реализация функции выборки из БД

Постановка задачи 

Выборка из файла с БД записей, в которых средний балл не ниже заданного граничного, и занесение этих записей в новый файл результатов выборки. Имена физических файлов с БД и для результатов выборки задает пользователь.

Метод решения

1) получить от пользователя имя физического файла с БД readln(namebd); 

2) если физический файл с таким именем отсутствует, то сообщить об этом пользователю и прекратить выполнение программы

3) получить от пользователя имя физического файла (namev) для хранения результатов выборки;

4) если файл с таким именем существует, то сообщить об этом пользователю и прекратить выполнение программы

5) получить от пользователя граничный средний балл (gsb)

6) связать логический и физический файлы с БД assign(fbd,namebd);
7) открыть файл прямого доступа с БД в режиме работы с существующим reset(fbd);
8) связать логический и физический файлы для результатов выборки assign(fv,namev);
9) открыть файл прямого доступа для результатов выборки в режиме создания нового файла
rewrite(fv);
10) перебрать все компоненты файла с БД. Пока не достигли конца файла  (not eof(fbd))

a) чтение очередного компонента БД read(fbd,r);

b) если r.sb

запись r в файл выборки write(fv,r);
11) закрыть файл с БД close(fbd);
12) закрыть файл с результатами выборки close(fv).
Данный метод справедлив и для последовательных файлов, так как никаких особенностей файлов прямого доступа не используется.

Информационная модель

Таблица 24.8. Информационная модель

	Статус
	Назначение
	Имя
	Тип

	Вход
	имя физического файла с БД
	namebd
	string

	Вход
	имя физического файла с результатами выборки
	namev
	string

	Вход
	граничный средний балл
	gsb
	real

	Пром
	имя логического файла с БД
	fbd
	tfileinfstud

	Пром
	имя логического файла с результатами выборки
	fv
	tfileinfstud

	Пром
	информация об учащемся
	r
	tinfstud


Программная модель

{программа выборки из файла по среднему баллу (файл Z4.PAS)}

program z4;

{$I FILE0.PAS}

var namev:string;{имя физического файла с выборкой}

    fv:tfileinfstud;{илф выборки}

    gsb:real;

begin

  writeln('Введите имя файла с информацией о студентах');

  readln(namebd);

  if not fs(namebd) then

    begin

      writeln('ОШИБКА!!! Файл с именем ',namebd,' не существует');

      halt

    end;

  writeln('Введите имя файла для выборки');

  readln(namev);

  if fs(namev) then

    begin

      writeln('ОШИБКА!!! Файл с именем ',namev,' существует');

      halt

    end;

  writeln('Граничный средний балл?');

  readln(gsb);

  assign(fbd,namebd);

  reset(fbd);

  assign(fv,namev);

  rewrite(fv);

  while not eof(fbd) do

        begin

          read(fbd,r);

          if r.sb>=gsb then

              write(fv,r);

         end;

  close(fbd);

  close(fv)

end.
Текст файла с общими описаниями

{файл с общими описаниями FILE0.PAS}

function fs(name:string):boolean;

var f:file;

begin

  assign(f,name);

  {$I-}

  reset(f);

  {$I+}

  if ioresult<>0 then

    fs:=false

  else

    begin

      fs:=true;

      close(f)

    end

end;

type tdata=record

       d:1..31;

       m:1..12;

       g:word

     end;

     tinfstud=record

       fam:string[20];

       dr:tdata;

       gp:word;

       sb:real

     end;

     tfileinfstud=file of tinfstud;

var fbd:tfileinfstud;

    namebd:string;

    r:tinfstud;

24.7. Организация работы с текстовыми файлами

Текстовый файл можно рассматривать как последовательный файл, но этот тип файлов из-за своих особенностей выделен в отдельный тип. 

Этот файл состоит из символов, среди которых могут встречаться специальные символы, называемые конец строки. Признак конец строки по разному хранится на разных внешних устройствах. Так на экране дисплея – это перевод курсора в начало следующей строки экрана, на клавиатуре – нажатие клавиши ENTER, на принтере – перевод печатающей головки в крайнее левое положение и прогон бумаги на одну строку вверх, на магнитном диске – это один или два машинных кода (в зависимости от операционной системы).

Таким образом, компонентами любого текстового файла могут быть: символы, признаки конец строки и признак конец файла. В Турбо Паскале для текстовых файлов определен стандартный тип TEXT. Этот тип автоматически определяется следующим образом

type text=file of char;

Для удобства текстовый файл можно рассматривать как последовательность позиций, в каждой из которых могут находиться любой из символов, признак конец строки и в последней позиции - признак конец файла. Для дальнейшего объяснения введем понятие текущая позиция текстового файла - позиция, с которой связан буфер файла. 

Буфер текстового файла, в отличие от других видов файлов, определяется при генерации операционной системы и имеет размер 128 или 255 байтов (Турбо Паскаль позволяет устанавливать размер буфера текстового файла с помощью процедуры SetTextBuf). 

При работе с текстовым файлом можно пользоваться следующими процедурами и функциями:

procedure reset(var <имя логического файла (в дальнейшем илф)>:text);

procedure rewrite(var <илф>:text);

procedure append(var <илф>:text);

procedure write(var <илф>:text;<записываемое в файл значение>:<имя типа для записи>);

procedure writeln(var <илф>:text);

procedure read(var <илф>:text;var <имя переменной>:<имя типа для чтения>);

procedure readln(var <илф>:text);

procedure flush(var <илф>:text);

function eof(var <илф>:text):boolean;

function eoln(var <илф>:text):boolean;

function seekeoln(var <илф>:text):boolean;

function seekeof(var <илф>:text):boolean;

Примечания: тип для операции записи – это любой целый, любой вещественный, логический boolean, литерный char, любой строковый тип;
тип для операции чтения – это любой целый, любой вещественный, литерный char, любой строковый тип;

Процедура записи в текстовый файл WRITE

Запись в текстовый файл значения в символьном виде осуществляется с помощью обращения к стандартной процедуре

write(<имя файловой переменной типа text>,<объект вывода>)

,где объект вывода определяется

<выражение>[:<минимальное количество позиций в символьном файле под значение выражения>
[:<количество позиций в строке символьного файла
 под дробную часть вещественного числа>]]

Из определения объекта вывода видно, что второй параметр процедуры write через двоеточие может сопровождаться дополнительной информацией – целым числом без знака типа byte. Это число определяет количество позиций в текстовом файле, отводимых под значение выражения, которое записывается в файл.

Результат выражения в объекте вывода может быть:

· любого из числовых типов;

· символьного типа char;

· любого из строковых типов string;

· логического типа boolean.

Правила выполнения обращения к процедуре write
Процедура write может быть выполнена только в том случае, если до выполнения процедуры буфер файла установлен на конец файла (для текстовых файлов это выполняется автоматически).

При выполнении процедуры write выполняются следующие действия:

1) - вычисляется значение выражения;

2) - осуществляется преобразования полученного значения в последовательность символов (не путать со строкой символов). Это не выполняется только для выражения типа CHAR;

3) - определяется количество позиций в символьном представлении значения;

4) - если при обращении к процедуре write второй параметр – выводимое выражение сопровождается через двоеточие дополнительной информацией -целым числом без знака, то возможно преобразование полученной последовательности символов. Это преобразование зависит от типа выражения, являющимся вторым параметром (правила преобразования указаны ниже);

5) - полученная последовательность символов заносится в буфер файла;

6) - как только буфер заполняется полностью, информация из буфера записывается в файл, начиная с позиции признака конец файла, с которым обязательно связан буфер до выполнения процедуры write. После последнего записанного в файл символа записывается признак конец файла. Буфер освобождается и связывается с позицией, в которой расположен этот признак. Так может повторяться до тех пор, пока вся полученная последовательность символов не занесется в буфер. Если буфер не заполнился полностью, то символы пока остаются в буфере.

Правила преобразования последовательности символов при записи в файл

Эти правила используются только в том случае, когда второй параметр при обращении к процедуре write сопровождается дополнительной информацией, и зависят от типа второго параметра.

Вариант 1: второй параметр любого целого типа

Если количество символов, полученных при преобразовании целочисленного значения в последовательность символов, меньше, чем число, указанн6ое в дополнительной информации, то слева к символам добавляются пробелы, так, чтобы количество полученных символов равнялось числу в дополнительной информации.

Вариант 2: второй параметр любого вещественного типа

Вещественное значение представляется в показательной форме с нормализованной мантиссой:

<знак числа><значащая цифра>.<цифры дробной части мантиссы>
e<знак порядка><целочисленный порядок>

Если математический сопроцессор не подключен, то дробная часть мантиссы содержит 10 цифр, целочисленный порядок – 2 цифры; при подключенном математическом сопроцессоре – соответственно 14 и 4 цифры.

В этом варианте результат преобразования последовательности символов зависит от настройки среды программирования (подключен или нет математический сопроцессор) и количества параметров в дополнительной информации (нет дополнительной информации, одно число в дополнительной информации или два числа):

a) если дополнительная информация отсутствует, то последовательность символов соответствует показательному (научному) формату числа, указанному в начале варианта (вместо знака числа плюс записывается пробел).

b) Если дополнительная информация состоит из одного числа, то это число может оказать влияние на количество позиций, отводимых в текстовом файле под символьное представление вещественного числа.

Если сопроцессор отсутствует и число в дополнительной информации меньше или равно 8, то последовательность символов преобразуется к следующему виду:

Zx.xEzxx
, где Z – знак числа (символ пробел, если знак плюс, символ -, если знак минус);

x – символ любой арабской цифры (первая цифра мантиссы всегда не 0);

z – знак порядка (символ + или символ -).

Если сопроцессор подключен и число в дополнительной информации меньше или равно 10, то последовательность символов преобразуется к следующему виду:

Zx.xEzxxxx

Если сопроцессор отсутствует и число в дополнительной информации ,больше 17, то слева к символам числа добавляются пробелы, так чтобы количество полученных символов равнялось дополнительной информации. В случае подключенного сопроцессора такие действия выполняются, когда число в дополнительной информации больше 23.

При всех других значениях числа в дополнительной информации отбрасываются лишние цифры в дробной части мантиссы (для числа 10 без сопроцессора последовательность символов не изменяется, это же выполняется при наличии сопроцессора, когда число в дополнительной информации 23).

c) если дополнительная информация состоит из двух чисел (второе число всегда меньше первого), последовательность символов преобразуется к представлению числа с фиксированной точкой, общепринятому в арифметике. Здесь второе число в дополнительной информации определяет количество позиций в дробной части записываемого в файл числа. А первое число задает общее количество позиций в записываемом числе, включая знак числа и десятичную точку. Если определяемое первым числом количество позиций не позволяет представить выводимое число, то это первое число не оказывает влияния на выделяемое в файле количество позиций под символьное представление записываемого числа. Если первое число в дополнительной информации позволяет представить в символьном виде записываемое число, то слева к символьному представлению могут добавляться пробелы, так чтобы количество позиций в последовательности символов равнялась первому числу в дополнительной информации.

Вариант 3: второй параметр типа CHAR

В этом случае к символу слева могут быть приписаны пробелы, так чтобы длина образованной последовательности символов равнялась числу в дополнительной информации.

Вариант 4: второй параметр любого строкового типа

Здесь число в дополнительной информации определяет минимальное количество позиций под символьное значение. Последовательность записываемых символов преобразуется по следующему правилу – если количество символов меньше, чем заданное число в дополнительной информации, то слева приписываются пробелы, так чтобы длина образованной последовательности символов равнялась числу в дополнительной информации.

Вариант 5: второй параметр типа boolean

В результате преобразования логического значения получается либо последовательность из четырех символов TRUE, либо последовательность из 5 символов FALSE. Дальнейшее преобразование этих последовательностей символов производится так же, как и в предыдущем варианте 4.

Процедура записи в текстовый файл WRITELN

Эта процедура заносит в буфер файла признак конец строки и записывает (выталкивает) содержимое буфера в текстовый файл, начиная с позиции признака конец файла. После последнего записанного в файл символа (это признак конец строки) записывается признак конец файла, буфер освобождается и связывается с этим признаком. 

Стандартное обращение к процедуре имеет вид:

writeln(<имя файловой переменной типа text>)

Пример записи в текстовый файл

program outfile;

var f:text;

begin

  assign(f,'c:\tp\user\dandatt.pas');

  rewrite(f);

  write(f,'пример':3);

  write(f,'вывода':10);

  write(f,true:6);

  write(f,false:1);

  writeln(f);

  write(f,-5:4);

  write(f,-252:1);

  write(f,32767);

  write(f,-3.56);

  writeln(f);

  write(f,567.42:8);

  writeln(f);

  write(f,567.42:1:4);

  close(f)

end.

В результате работы программы будет создан файл c:\tp\user\dandatt.pas. Содержимое этого файла приведено на рис. 24.4 (компоненты на рисунке располагаются слева направо, сверху вниз).

Рис. 24.4. – Содержимое файла

Процедура чтения из текстового файла READ

Чтение из текстового файла осуществляется с помощью обращения к стандартной процедуре

read(<илф типа text>, <объект ввода>)

, где объект ввода это имя простой переменной:

- любого из числовых типов;

- символьного типа char;

- любого из строковых типов string

или имя простого компонента переменной структурированного типа.

Правила выполнения процедуры read
Синтаксическое ограничение при вызове процедуры: второй параметр обязательно любого целого, любого вещественного, литерного char или любого строкового типа. Правила выполнения зависят от типа второго параметра.

Вариант 1: второй параметр числового типа
 (любого целого, либо любого вещественного)

1) В буфер файла считываются позиции текстового файла, начиная с позиции, с которой был связан буфер до обращения к процедуре read до первой позиции, в которой располагается либо пробел, либо символ табуляции, либо признак конец строки, либо признак конец файла. При этом начальные позиции с пробелами, символами табуляции и признаками конец строки игнорируются.

2) Прочитанные в буфер символы преобразуются в значение типа данных второго параметра. Если из файла не прочитано ни одного символа (буфер связан с признаком конец файла), то считается, что прочитан 0.

3) Если при преобразовании встретился символ, который не может быть преобразован в численное значение, то возникает ошибка ввода-вывода.

4) Если в результате преобразования получено число, не принадлежащее множеству значений типа второго параметра, то возникает ошибка выхода из диапазона значений.

5) При успешном преобразовании полученное значение заносится в оперативную память под именем второго параметра. Буфер файла освобождается. Новое положение буфера связано с позицией файла, в которой прекращено чтение (это либо пробел, либо признак конец строки, либо признак конец файла).

Вариант 2: второй параметр литерного типа char
1) В буфер считывается одна позиция текстового файла. Если это не признак конец файла, то буфер связывается со следующей позицией текстового файла. Если до выполнения процедуры read был связан с признаком конец файла, то местоположение буфера не изменяется.

2) Прочитанный символ заносится в оперативную память под именем второго параметра.

Вариант 3: второй параметр любого строкового типа

1) В буфер файла считывается столько позиций из текстового файла, сколько определено в строковом типе для второго параметра, если ранее не встретился признак конца строки или признак конца файла.

2) Если встретился признак конец строки или конец файла, то буфер связывается с позицией признака. Если при чтении эти признаки не встретились, то новое положение буфера – первая непрочитанная позиция.

3) Информация из буфера заносится в оперативную память под именем второго параметра, при этом формируется динамическая длина прочитанной строки.

Процедура пропуска очередного признака конец строки READLN

После выполнения обращения к процедуре readln(<илф типа text>)
буфер связывается со следующей позицией за ближайшим справа признаком конец строки. Если этот признак не встретился, то буфер связывается с признаком конец строки.
Пример чтения из текстового файла

Пусть в текстовом файле c:\tp\user\inp.pas находится информация, приведенная на рис. 24.5.

Рис. 24.5. – Содержимое текстового файла

При выполнении программы:

program inptxt;

var f:text;

    a:integer;

    b:real;

    c:char;

    d:string[6];

    e:string;

begin

  assign(f,'c:\tp\user\inp.pas');

  reset(f);

  read(f,c);

  read(f,a);

  read(f,b);

  readln(f);

  read(f,d);

  read(f,e);

  close(f)

end.

сформируются значения переменных, приведенные на рис. 24.6.

	A
	5

	B
	7.5e-3

	C
	'F'

	D
	'ПРИМЕР'

	E
	'ЧТЕНИЯ ИЗ ФАЙЛА'


Рис. 24.6. – Содержимое ОП

Функции проверки местонахождения буфера

function EOF(var <илф>:text):Boolean;

возвращает TRUE, если буфер связан с признаком конец файла, в противном случае –FALSE.

function EOLn(var <илф>:text):Boolean;

возвращает TRUE, если буфер связан признаком конец строки или с признаком конец файла, в противном случае –FALSE.

function SeekEOF(var <илф>:text):Boolean;

возвращает TRUE, если буфер связан с признаком конец файла или перед этим признаком располагаются только пробелы, признаки конец строки, символы табуляции, в противном случае –FALSE.

function SeekEOLn(var <илф>:text):Boolean;

возвращает TRUE, если буфер связан с признаком конец строки или с признаком конец файла или перед этим признаком располагаются только пробелы, признаки конец строки, символы табуляции, в противном случае –FALSE.

Процедура записи содержимого буфера в файл

procedure flush(var <илф>:text); 

эта процедура выводит текущее содержимое буфера файла, не дожидаясь заполнения буфера до конца.

24.8. Стандартные текстовые файлы INPUT, OUTPUT
В каждой программе автоматически используются два текстовых файла input и output. Логический файл input – стандартный файл для ввода информации. Для ПЭВМ ему соответствует физический файл, расположенный на клавиатуре. Логический файл output – стандартный файл для вывода информации. Для ПЭВМ ему соответствует физический файл, расположенный на экране дисплея.

В каждую программу компилятор добавляет следующие определения и действия:

a) в раздел описаний

type text=file of char;

var input, output : text;

б) в начало раздела операторов

assign(input, 'con');

reset(f);

assign(output, 'con');

rewrite(output);

в) в конец раздела операторов 

close(input);

close(output);

г) во все обращения к стандартным процедурам ввода-вывола, в которых отсутствует первый параметр – имя логического файла, вставляется этот первый параметр, для процедур ввода это input, для процедур вывода – output.

24.9. Обработка компилятором обращений к процедурам чтения и записи

Необходимо помнить, что стандартные обращения к процедурам read и write имеют только два параметра, причем первый это всегда имя логического файла. Если в обращении отсутствует этот первый параметр, то автоматически подставляется либо имя input, либо – output. В свою очередь стандартные процедуры readln и writeln имеют только один параметр – имя логического файла типа text. Если этот параметр отсутствует, то соответственно подставляется input или output.

Любое обращение в программе к процедурам ввода-вывода приводится к стандартному виду по следующему правилу:

1) Обращение вида readln (<p1>,<p2>,<p3>,   ,<pn>)
где pi – имя i-ой переменной, заменяется компилятором следующим составным оператором

begin


read(input, <p1>;


read(input, <p2>);


read(input, <p3>);


...


read(input, <pn>);


readln(input)

end;

2) Обращение вида writeln (<v1>,<v2>,<v3>,   ,<vn>)
где vi – i-ый объект вывода, заменяется компилятором следующим составным оператором

begin


write(output, <v1>;


write(output, <v2>);


write(output, <v3>);


...


write(output, <vn>);


writeln(output)

end;

3) если имя логического файла указано в обращении к процедуре, то это имя используется в составном операторе в стандартных обращениях к соответствующим процедурам ввода-вывода.

24.10. Общие процедуры работы с файлами и каталогами

Эти процедуры зависят от операционной системы, для которой создаются программы.

procedure Rename(var <имя логического файла>:<любой файловый тип>;<новое имя физического файла>:string);

переименовывает физический файл ранее связанный с заданным именем логического файла (применяется для закрытых файлов).

procedure Erase(var <имя логического файла>:<любой файловый тип>);

уничтожает физический файл, связанный с заданным именем логического файла (файл должен быть закрыт).

procedure GetDir(<номер дисковода>:byte; var <имя текущего каталога>:string);

определяет имя текущего каталога для задаваемого номером дисковода (0 – текущий диск; 1 – диск А; 2 – диск В; 3 – диск С и т.д.). Результат располагается во втором параметре.

procedure ChDir(<имя текущего каталога>:string);

устанавливает текущим каталог с заданным именем.

procedure MkDir(<имя каталога>:string);

создает новый каталог с указанным именем

procedure RmDir(<имя каталога>:string);

удаляет пустой не текущий каталог с указанным именем.

